Continuation-based Partial Evaluation Without Continuations

Peter Thiemann thiemann@informatik.uni-freiburg.de

Universität Freiburg, Germany

Abstract. Continuation-based partial evaluation is an implementation technique for program specializers that allows generated code contexts to commute with specialization-time values. Thus it enables additional specialization-time computations. As the name suggests, a popular implementation is based on the manipulation of continuations. We show that a substitution monad provides an alternative implementation of the context manipulation performed by continuation-based partial evaluation. We prove its equivalence to the original continuation-based implementation. Our new formalization allows for a more precise bindingtime analysis that is based on a type and effect system. We prove the soundness of the type and effect system with respect to our implementation and point out how a binding-time analysis based on this system may avoid code duplication and enable the generation of more efficient generating extensions.

Key Words: functional programming, partial evaluation, type systems

1 Introduction

Partial evaluation [4,15] is an automatic program transformation that specializes a program with respect to known input data. Of particular interest is offlinepartial evaluation, where each part of the program is annotated as either executable at specialization time or as generating specialized code. For example, in the program

power x n = if^s n ==^s 0 then (lift 1) else x ∗^d power x (n −^s 1)

the superscript s (static) indicates a specialization-time expression, whereas the superscript d (dynamic) indicates a run-time expression. The function lift converts the internal representation of a number into a constant expression (a syntax tree). In the example, running the program power x 2 yields the syntax tree

x * (x * 1)

The annotations may be added to the program manually or by means of a program analysis (binding-time analysis). In either case, a type system can be used to specify when an annotation is sound.

The binding-time analysis is an asset and a drawback at the same time. It is an asset because it clearly indicates which expressions are processed at specialization time (and which are not). It is also a drawback because some programs require non-trivial transformation (binding-time improvements) before they specialize well. Hence, some research has been geared towards automatizing these transformations and including them into partial evaluation systems [1, 7, 8, 16, 20].

The present work concerns continuation-based partial evaluation, which incorporates one of the most essential binding-time improvements for call-by-value functional programming languages. It was pioneered by Bondorf, Consel, and Danvy [1, 3] and later refined and explained by Danvy, Hatcliff, Lawall, and the present author [13, 16, 17]. The basic idea is that code may be generated interleaved with specialization-time computations. For example

lift (17 +^s let^d x = e in 4)

specializes to

let x = e in 21

that is, the code generation for the surrounding let x = e in . . . is interleaved with the specialization-time computation 17 + 4. It turns out [13, 17], that this feature is essential to achieve sound specialization for languages with effects.

Danvy and Lawall [16] have proposed to extend interleaved code generation to also include dynamic conditionals. Extending the previous example

lift (17 +^s if0^d e 4 25)

specializes to

if0 e 21 42

that is, the computation (17 + [ ]) is performed twice and the conditional floats to the top level of the expression. This feature vastly simplifies “The Trick”, a common binding-time improvement [7].

As the name “continuation-based partial evaluation” suggests, the first implementation of interleaved code generation was in terms of a non-standard use of continuations, either in the source program [3] or in the implementation of the specializer [1, 16]. Later on, further state-based implementations have been suggested [11, 23].

In the present work, we start from a monadic specification for continuationbased partial evaluation and suggest an implementation of the underlying monad that does not rely on continuations. We show that our new implementation is equivalent to an implementation with continuations and develop a novel annotated type and effect system for two-level call-by-value lambda calculus that models interleaved code generation (as demonstrated with the let and if above) as a specialization-time effect. This type system may be used to optimize specialization by taking advantage of (the absence of) specialization-time effects.

Terms e ::= x | λ^bx.e | e@^be | let^b x = e in e | c | succ^b e | if0^b e e e | lift e | reset e Binding times b ::= s | d s < d Types τ ::= int | ρ^t−→ρ Annotated types ρ ::= (τ, b) Effects t ::= 1 | i(t, t) | l(t) | t ≫ t | t ⊔ t

Fig. 1. Two-level language

Overview In Sec. 2, we present a call-by-value two-level lambda calculus and a type and effect system specifying its static semantics. Section 3 specifies its dynamic semantics via a translation into the computational metalanguage augmented with code-generating operators. The translation requires some operations from the underlying monad and Sec. 4 states these requirements in the form of equational axioms and provides two implementations for the monad, one with continuations and another with substitutions. Section 5 argues informally that the two monads are equivalent. In Sec. 6, we prove type soundness for the two-level lambda calculus and in Sec. 7 we give some examples, how specialization may be optimized with the presented system. Section 8 discusses some related work and Section 9 concludes.

2 Two-level lambda calculus

We consider an applied two-level call-by-value lambda calculus with built-in natural numbers. Its syntax is given in Fig. 1. Each subexpression except for variables, constants, left and reset expressions is annotated with a bindingtime annotation b, which can be s (static) or d (dynamic). The computation rules are beta value reduction for static beta redexes and static let expressions, and delta reduction for static successors succ and testing for zero if0. The only non-standard expression is reset e and we will defer its explanation to Sec. 7.

The syntax of two-level types in Fig. 1 is standard [9], except for the addition of effects. Effects characterize the interleaved code generation as exemplified by the let x = e in . . . in the introduction. The effect 1 indicates that no such code is generated, l(t) indicates that a let x = e in e^′ is generated where e and x are arbitrary and e^′ is described by t, i(t_1, t_2) indicates an interleaved if0 e e_1 e_2 with arbitrary e but branches e_1 and e_2 specified by t_1 and t_2, respectively. Of the remaining operations, effect union, t_1 ⊔t_2, specifies interleaved code described by either t_1 or t_2 and t_1 ≫ t_2 specifies sequential composition. Fig. 2 specifies these operations on effects. In essence, each effect can be simplified to a union of simple effects generated by 1, l(t), and i(t^′, t^{′′}). The union operator can be floated to the top level using equations (8), (9), (10), and (11). Sequential composition ≫ can be pushed through towards 1s using the equations (5), (6), and (7) and then t ⊔ t = t (1) t ⊔ t^′ = t^′ ⊔ t (2) t ⊔ (t^′ ⊔ t^{′′}) = (t ⊔ t^′) ⊔ t^{′′} (3) 1 ≫ t = t (4) l(t^′) ≫ t = l(t^′ ≫ t) (5) i(t^′, t^{′′}) ≫ t = i((t^′ ≫ t), (t^{′′} ≫ t)) (6) (t^′ ≫ t^{′′}) ≫ t = t^′ ≫ (t^{′′} ≫ t) (7) (t^′ ⊔ t^{′′}) ≫ t = (t^′ ≫ t) ⊔ (t^{′′} ≫ t) (8) l(t^′ ⊔ t^{′′}) = l(t^′) ⊔ l(t^{′′}) (9) i(t^′ ⊔ t^{′′}, t) = i(t^′, t) ⊔ i(t^{′′}, t) (10) i(t, t^′ ⊔ t^{′′}) = i(t, t^′) ⊔ i(t, t^{′′}) (11)

Fig. 2. Algebraic properties of effects

eliminated by (4)^1. The type and effect system presented in Fig. 3 for specifying binding times is new. It relies on the inference system in Fig. 4 to specify a wellformedness criterion for types. The typing part of the system is standard, the new ingredient is the effect part.

To illustrate the operations on effects, let us consider the rule for function application. It states that if specialization of e_1 and e_2 (in that order) generates interleaved code as described by t_1 and t_2, and if applying the value of e_1 to the value of e_2 generates interleaved code as described by the latent effect t, then the interleaved code generated by e_1@^be_2 starts as described by t_1, followed by t_2, and finally t. Similar reasoning applies to the static and the dynamic let expressions. In the latter, the explicit l(. . .) stands for the let created by the expression itself (which is always interleaved, so that it can float out of static contexts). Since the outcome of the static conditional cannot be determined before specialization, the interleaved code is approximated by taking the union of the two effects. The dynamic conditional also generates its code interleaved by default and hence need not constrain the binding time of its branches^2.

The reset · expression is also a novelty of our presentation. As its typing rule specifies, it is applicable to any dynamic value (a syntax tree) and is used to flush the interleaved code by dumping it on top of that syntax tree. Its effect is empty, 1, since all interleaved code is now part of the dynamic value.

As usual in an effect system [18], each function type carries a latent effect indicating the effect that is unleashed when the function is applied. The wellformedness judgement enforces the usual constraints on the binding-time an-

1 This elimination relies on the fact that there are no effect variables in the system. 2 A thorough discussion of this feature is out of the scope of this paper but may be found in the literature [7, 16].

⊢wft ρ Γ, x : ρ ⊢ x : ρ ! 1 Γ ⊢ c : (int, s) ! 1

Γ ⊢ e : (int, b) ! t Γ ⊢ succ^b e : (int, b) ! t

Γ ⊢ e : (int, s) ! t Γ ⊢ lift e : (int, d) ! t Γ, x : ρ2 ⊢ e : ρ1 ! t ρ = (ρ2^t−→ρ1, b) ⊢wft ρ Γ ⊢ λ^bx.e : ρ ! 1 Γ ⊢ e1 : (ρ2^t−→ρ1, b) ! t1 Γ ⊢ e2 : ρ2 ! t2 Γ ⊢ e1@^be2 : ρ1 ! t1 ≫ t2 ≫ t Γ ⊢ e1 : (τ, s) ! t1 Γ, x : (τ, s) ⊢ e2 : ρ2 ! t2 Γ ⊢ let^s x = e1 in e2 : ρ2 ! t1 ≫ t2 Γ ⊢ e1 : (τ, d) ! t1 Γ, x : (τ, d) ⊢ e2 : ρ2 ! t2 Γ ⊢ let^d x = e1 in e2 : ρ2 ! t1 ≫ l(t2) Γ ⊢ e1 : (int, s) ! t1 Γ ⊢ e2 : ρ ! t2 Γ ⊢ e3 : ρ ! t3 Γ ⊢ if0^s e1 e2 e3 : ρ ! t1 ≫ (t2 ⊔ t3) Γ ⊢ e1 : (int, d) ! t1 Γ ⊢ e2 : ρ ! t2 Γ ⊢ e3 : ρ ! t3 Γ ⊢ if0^d e1 e2 e3 : ρ ! t1 ≫ i(t2, t3) Γ ⊢ e : (τ, d) ! t Γ ⊢ reset e : (τ, d) ! 1

Fig. 3. Typing rules

⊢wft (int, b) ⊢wft (τ2, b2) ⊢wft (τ1, b1) b ≤ b1 b ≤ b2 if b = d then t = 1 ⊢wft ((τ2, b2)^t−→(τ1, b1), b)

Fig. 4. Wellformedness of annotated types

T �x� = return x T �λ^sx.e� = return (λx.T �e�) T �λ^dx.e� = newVar >>= λx.T �e� >>= λe^′.return (LamExp x e^′) T �e1@^se2� = T �e1� >>= λf.T �e2� >>= λv.f@v T �e1@^de2� = T �e1� >>= λe^′_1.T �e2� >>= λe^′_2.return (AppExp e^′_1 e^′_2) T �let^s x = e1 in e2� = T �e1� >>= λx.T �e2� T �let^d x = e1 in e2� = T �e1� >>= λe^′_1.newVar >>= λx.mlet x e^′_1 T �e2� T �c� = return c T �succ^s e� = T �e� >>= λv.return (succ v) T �succ^d e� = T �e� >>= λe^′.return (SuccExp e^′) T �if0^s e1 e2 e3� = T �e1� >>= λv1.if0 v1 T �e2� T �e3� T �if0^d e1 e2 e3� = T �e1� >>= λe^′_1.mif e^′_1 T �e2� T �e3� T �lift e� = T �e� >>= λv.return (ConExp v) T �reset e� = mreset T �e�

Fig. 5. Translation into augmented metalanguage

notations. In addition, it ensures that the application of a dynamic (run-time) function does not generate interleaved code.

3 Monadic specification of partial evaluation

Effects have been abstracted and described using monads [10, 25]. For that reason, we construct a monadic specification of partial evaluation, using special operators that exert the effects, mlet for l(t) and mif for i(t, t), or flush them, mreset. There is also a newVar operator for creating fresh variable names, which is left unspecified to avoid further complication.

We specify the semantics of the two-level calculus by translation into a variant of Moggi’s computational metalanguage [21], that is, the call-by-name lambda calculus augmented with the monadic return · and bind operations (·>>=·) plus the special newVar, mlet, mif, and mreset operators. There are also constructors for generating syntax trees for applied call-by-value lambda calculus.

For concreteness, we specify syntax trees and the operations in the form of Haskell definitions [12]. Here is the datatype for syntax trees.

type Var = String
data Exp =
    VarExp Var -- variable
  | ConExp String -- constant
  | SuccExp Exp -- successor
  | LamExp Var Exp -- lambda
  | AppExp Exp Exp -- application
  | LetExp Var Exp Exp -- let
  | IfExp Exp Exp Exp -- conditional

We capture the further requirements in a type class PEMonad, which states the required operations on a monad m, suitable for implementing partial evaluation, and their types.

class Monad m => PEMonad m where
  mlet :: Var -> Exp -> m a -> m a
  mif :: Exp -> m a -> m a -> m a
  mreset :: m Exp -> m Exp

The type signatures clearly show that

– mlet transforms a computation (its last argument) into another computation;

– mif transforms two computations into another computation; – the type (and hence the binding time) of the body of the mlet as well as the type of the branches of the mif is unconstrained;

– mreset mreset requires that its argument computation returns a syntax tree (i.e., a dynamic value) and it returns a syntax tree, too.

The translation rules in Fig. 5 are reasonably straightforward. For the static constructs they correspond exactly to the standard monadic translation. As an example for a dynamic construct, we consider the translation of the dynamic lambda:

T �λ^dx.e� = newVar >>= λx.T �e� >>= λe^′.return (LamExp x e^′)

The translated term creates a fresh variable name and binds it to x, the bound variable of the translated lambda. In that environment, it specializes e, binds the result to e^′, and returns the syntax tree LamExp x e^′. For understanding the rules, it is important to keep in mind that T �·� is a translation, that is, the x and e on both sides are identical.

To avoid clutter in the translation (and in the proofs below) we assume that the type Var is a subset of the type Exp with VarExp being the injection function. The type-correct right-hand side for the dynamic lambda would be

newVar >>= λx^′.let x = VarExp x^′ in T �e� >>= λe^′.return (LamExp x^′ e^′)

and similarly in the translation of the dynamic let.

4 Monads for partial evaluation

Having constructed the monadic translation for the two-level lambda calculus, the next task is to define a suitable monad that supports the operations, mlet, mif, and mreset.^3 In fact, we will present two such monads. One of them is the continuation monad and its definitions correspond to the usual implementation

3 Recall that we leave newVar unspecified, although it is straightforward to add.

of continuation-based partial evaluation [1,16]. The other monad is, perhaps surprisingly, a simple substitution monad. Nevertheless, as we will show in Sec. 4.3, both monads fulfill the expected laws for the new operations and we will use these laws in Sec. 5 to prove that the results of interpreting a translated twolevel term in either monad are equivalent.

4.1 Continuation monad

The type for the continuation monad is standard. Since we are defining a monad for partial evaluation, where the result is always a syntax tree, the type of answers is Exp. For technical reasons, Haskell requires the definition of a new data type

C and a corresponding untagging operation unC.

data C a = C ((a -> Exp) -> Exp)
unC (C k) = k

The definitions for return and >>= are standard^4.

instance Monad C where
  return x = C (\ c -> c x)
  ma >>= f = C (\ c -> unC ma (\ x -> unC (f x) c))

Of course, the essence of continuation-based partial evaluation lies in the defintions of the PEMonad operators. The definitions below correspond to code fragments from Bondorf, Danvy, and Lawall [1, 16]. In particular, note how the continuation is duplicated in the mif.

instance PEMonad C where
  mlet v e ma =
    C (\ c -> LetExp v e (unC ma c))
  mif e m1 m2 =
    C (\ c -> IfExp e (unC m1 c) (unC m2 c))
  mreset me =
    C (\ c -> c (unC me id))

4.2 Substitution monad

A computation of type S a in the substitution monad is a tree where the leaves hold the computed values of type a. Each inner nodes is either an L node, specifying an interleaved let expression, or an I node, specifying an interleaved if expression. The following definition captures this intuition.

data S a = V a
         | L Var Exp (S a)
         | I Exp (S a) (S a)

4 Recall that \ x -> e is a lambda abstraction.

The monadic operations are simple to specify. The return operation creates a trivial tree using the V constructor. The bind operation m >>= f propagates f to each leaf of m, applies it to the value x found there, and pastes the resulting tree f x in place of the leaf.

instance Monad S where
  return x = V x
  ma >>= f = case ma of
               V x -> f x
               L v e s -> L v e (s >>= f)
               I b s1 s2 -> I b (s1 >>= f) (s2 >>= f)

The operations of the PEMonad class are straightforward to implement, too. The interleaved code can be prepended to the tree since all operations are applied to the values in the leaves of the tree, as specified by the >>= operation. The

mreset operation collects the tree and translates it into an expression.

instance PEMonad S where
  mlet v e ma = L v e ma
  mif e m1 m2 = I e m1 m2
  mreset me =
    case me of
      V e -> return e
      L v e me -> mreset me >>= \ e’ -> return (LetExp v e e’)
      I b m1 m2 -> mreset m1 >>= \ e1 -> mreset m2 >>= \ e2 ->
                   return (IfExp b e1 e2)

4.3 Laws

Besides fulfilling the usual monad laws [26], a PEMonad must obey a number of additional laws. These laws characterize the interaction of mlet and mif with the >>= operator, as well as the interaction of mlet and mif with mreset.

The first two laws state that the mlet and mif operations are both associative with respect to >>=.

(MLB) mlet v e ma >>= f == mlet v e (ma >>= f) (MIB) mif e m1 m2 >>= f == mif e (m1 >>= f) (m2 >>= f)

The next three laws state that the mreset operation absorbs an mlet or an

mif at the price of generating the corresponding code.

     (MRR) mreset (return e) == return e
     (MRL) mreset (mlet v1 e1 me) == mreset me >>= λ e ->
                                      return (LetExp v1 e1 e)
     (MRI) mreset (mif e m1 m2) == mreset m1 >>= λ e1 ->
                                      mreset m2 >>= λ e2 ->
                                      return (IfExp e e1 e2)

Lemma 1. The laws (MLB), (MIB), (MRR), (MRL), and (MRI) hold in C and S.

Proof. See Appendix A.

In addition, the mlet and mif operations for a PEMonad obey the monadic flattening laws on the level of generated syntax trees. To state these laws, we define two auxiliary operations, mmlet and mmif, that incorporate the computation of their first expression:

mmlet :: Var -> m Exp -> m a -> m a
mmif :: m Exp -> m a -> m a -> m a

mmlet v me ma = me >>= \ e -> mlet v e ma
mmif me m1 m2 = me >>= \ e -> mif e m1 m2

Using these definitions, we can state flattening laws corresponding to the third monad law (associativity of let/bind), on the level of interleaved generated code.

mmlet v2 (mmlet v1 m1 m2) m3 == mmlet v1 m1 (mmlet v2 m2 m3)
mmlet v (mmif m1 m2 m3) m4 == mmif m1 (mmlet v m2 m4) (mmlet v m3 m4)
mmif (mmlet v m1 m2) m3 m4 == mmlet v m1 (mmif m2 m3 m4)
mmif (mmif m1 m2 m3) m4 m5 == mmif m1 (mmif m2 m4 m5) (mmif m3 m4 m5)

They also hold for both, S and C, and they are proven using (MLB) and (MIB).

4.4 New variables

As stated before, we are not going to specify the newVar operation completely. However, we have to make some assumptions about it. Its type is

newVar :: PEMonad m => m Var

and its only effect is the generation of a fresh variable name. That is,

newVar == (νx ∈ Var) return x

5 Equivalence of PE monads

It turns out that the laws (MLB), (MIB), (MRR), (MRL), and (MRI) completely characterize the computational behavior of the operations of PEMonad. To make that formal, let λ_{P E} be the equational theory of the computational metalanguage (beta equivalence plus the three monad laws) augmented by delta rules for successor and if0 and the mentioned PEMonad laws.

We take equivalence of the two PEMonads introduced above to mean that, for each closed two-level lambda term e of type (τ, d), “running” the term T �e� yields the same answer. Here are the definitions for running C and S computations.

runC :: C Exp -> Exp
runC me = unC (mreset me) id

runS :: S Exp -> Exp
runS se = let V x = mreset se in x

To (informally) demonstrate equivalence in λ_{P E} , observe that the laws (MLB) and (MIB) allow mlet and mif operations to float outside of >>= operations. That means, each translated term is equal to one where all the mlets and mifs, interspersed with newVars, are aggregated either at the top of the term or in the argument of mresets. After application of mreset at the toplevel, in runC or runS, and performing newVars, the laws (MRR), (MRL), and (MRI) enable all mlets and mifs to be removed. The resulting term does not contain monadic operators except return and >>=, hence its computed value is independent of the underlying monad.

6 Soundness

In this section, we connect the type system with the semantics. We shall argue in denotational style by first defining a semantics of a two-level type with effect and then proving type soundness by examining the translated term in the S monad. Since the simply-typed lambda calculus, the basis of our two-level calculus, is strongly normalizing, we state the semantics in terms of sets. In addition, we assume a standard semantics mapping �·� that maps an expression, e, and a value environment, φ, to a value (see for example [19]). We make this more formal by defining the carriers of an appropriate Henkin model for the type language of the metalanguage.

Definition 1 (Types of the metalanguage).

σ ::= Var | Exp | M σ | σ -> σ

Definition 2 (Semantic values). Let �Var� be the set of variable names.

Let �Exp� be the set of expressions generated by the definition of Exp in Sec. 4. Let �Int� be the set of natural numbers. For each type σ, let �M σ� = {V v | v ∈ �σ�} ∪ {L x e m | x ∈ �Var�, e ∈ �Exp�, m ∈ �M σ�} ∪ {I e m_1 m_2 | e ∈ �Exp�, m_i ∈ �M σ�}

For each types σ_2 and σ_1, let

�σ_2 -> σ_1� = �σ_1�^{�σ2 �}

that is, the set theoretic functions from �σ_2� to �σ_1�.

Clearly, this family of carrier sets with set-theoretic function application satisfies the preconditions for a Henkin model. Now, we can define the semantics of an annotated type in terms of the Henkin model.

Definition 3 (Semantics of annotated types).

�ρ ! 1� = {V v | v ∈ �ρ�} �ρ ! l(t)� = {L x e v | x ∈ �Var�, e ∈ �Exp�, v ∈ �ρ ! t�} �ρ ! i(t_1, t_2)� = {I e v_1 v_2 | e ∈ �Exp�, v_i ∈ �ρ ! t_i�} �ρ ! t_1 ⊔ t_2� = �ρ ! t_1� ∪ �ρ ! t_1� �(τ, d)� = �Exp� �(int, s)� = �Int� �(ρ_2^t−→ρ_1, s)� = {f | (∀x ∈ �ρ_2�) f x ∈ �ρ_1 ! t�}

It turns out that the translation T �·� is type preserving.

Theorem 1 (Translation soundness). Suppose that Γ ⊢ e : ρ ! t.

Then T �Γ � ⊢ T �e� : M T �ρ� in the simply-typed lambda calculus where

– M is a PEMonad (for instance, S or C); – T �Γ �(x) = T �Γ (x)�, for all x ∈ dom(Γ ); – T �(τ, d)� = Exp; – T �(int, s)� = Int;

– T �(ρ_2^t−→ρ_1, s)� = T �ρ_2� -> M T �ρ_1�.

In other words, Theorem 1 establishes that each well-typed two-level lambda expression may be translated into a standard functional programming language (e.g., Haskell or ML) and executed there to achieve specialization. Hence, the translation T �·� creates a generating extension, i.e., a dedicated specializer for one particular two-level lambda expression. See the author’s article [24] for more background information.

To prove type and effect soundness, we need an auxiliary lemma. It establishes that all types inferred by the inference system are well-formed.

Lemma 2. Suppose that Γ ⊢ e : ρ ! t and (∀x ∈ dom(Γ )) ⊢wft Γ (x).

Then ⊢wft ρ.

Proof. Standard induction on the derivation of Γ ⊢ e : ρ ! t.

Theorem 2 (Type and effect soundness). Suppose that Γ ⊢ e : ρ ! t and, for all x ∈ dom(Γ ), φ(x) ∈ �Γ (x)� and ⊢wft Γ (x).

Then �T �e�φ ∈ �ρ ! t�.

Proof. See Appendix B

Adding recursion to our framework would give rise to two complications. First, we would have to base the semantics on domains instead of sets. Second, we would have to add recursion in the form μe.t to the language of effects. Both are standard [19, 22] and are elided here for the sake of conciseness.

7 Optimized specification

The two-level calculus presented in Sec. 2 has one unusual feature compared to standard accounts (e.g., in the text book [15] or in the papers about continuationbased partial evaluation [1, 16]). The expression reset e is usually not made explicit in the two-level language, but rather hidden in the translation of the dynamic versions of the binding constructs, let^d x = e in . . . and λ^dx. . . ..

With our formal system, the reset e expression plays a clearly defined role. It coerces a dynamic two-level term with arbitrary effect into a two-level term without effect. Instead of integrating reset · into the translation of λ^dx.e, the type system makes sure that the effect of e is 1 (that is, empty). This way, the binding-time analysis is free to place reset · as early as possible (this avoids code duplication in specialized expressions and hence more efficient specialization) or to avoid placing reset · at all, if the type system infers that the effect is 1, already.

Here are two examples that illustrate either point.

– Code duplication. Let Γ (f_1) = Γ (f_2) = (ρ_2^1−→ρ_1, d), Γ (x) = (int, d), and Γ (a) = ρ_2. Then

Γ ⊢ (if0^d x f_1 f_2)@^da : ρ_1 ! i(1, 1)

Specializing this code as it is would lead to duplicating the application and its argument a:

if0^d x (f_1@^da) (f_2@^da)

However, since both f_1 and f_2 are dynamic, the binding-time analysis may wrap the conditional into a reset · so that the effect of the term vanishes:

Γ ⊢ (reset (if0^d x f_1 f_2))@^da : ρ_1 ! 1

Specializing this code avoids the duplication:

(if0 x f_1 f_2)@a

– Avoiding the placement of reset ·. Suppose that

Γ, x : ρ_2 ⊢ e : ρ_1 ! 1

where ⊢wft (ρ_2^1−→ρ_1, d) so that

Γ ⊢ λ^dx.e : (ρ_2^1−→ρ_1, d) ! 1

Observe that it is not necessary to insert a reset · on top of the lambda’s body because the body’s effect is empty. In constrast, a conventional continuationbased partial evaluator always insert a reset · in this place.

8 Related work

Consel and Danvy [3] were the first to argue that the binding time of dynamic let expressions can be improved converting source programs to continuationpassing style.

Subsequently, Bondorf [1] discovered that the same binding-time improvement can be achieved by writing the partial evaluator in (not-quite) continuationpassing style. In fact, it is essential to deviate from standard continuation-passing style in the treatment of the dynamic let and lambda expressions.

Danvy and Lawall [16] showed that the non-standard handling of continuations in Bondorf’s partial evaluator can be implemented using the control operators shift and reset [6] that provide compositional continuations. They also generalize the treatment of the dynamic let to the dynamic if.

It was later seen that the treatment of dynamic let expressions (first termed “let-insertion” [2]) enabled by continuation-based partial evaluation was essential to achieve a satisfactory and sound specialization for languages with computational effects [13, 17].

Recently, the work on type-directed partial evaluation for call-by-value languages [5, 11] has spread new interest on the implementation of let-insertion. In addition, work to define powerful, efficient online program generators has lead to the discovery of a state-based implementation of let-insertion [23]. The latter work also proves the equivalence of their let-insertion to the continuation-based method using a type-indexed relation. However, their let-insertion does not scale to if-insertion in an obvious way and our correspondence is much more direct.

9 Conclusion

This work has two major contributions.

1. It presents a new and simple formalization and implementation of continuationbased partial evaluation. This formalization is pleasingly easy to reason about and it is proven equivalent to the traditional one with continuations.

2. It presents a new type and effect system for two-level lambda calculus. This type system can form the basis for a binding-time analysis that constructs optimized generating extensions.

In future work, we would like to examine the connection to the state-based formulation of continuation-based partial evaluation and we would like to design and implement an actual binding-time analysis based on the type system presented in this work. It would also be interesting to compare the effiency of our substitution-based implementation with the ones using continuations and state.

References

1. Anders Bondorf. Improving binding times without explicit CPS-conversion. In Proc. 1992 ACM Conference on Lisp and Functional Programming, pages 1–10, San Francisco, California, USA, June 1992.

2. Anders Bondorf and Olivier Danvy. Automatic autoprojection of recursive equations with global variables and abstract data types. Science of Computer Programming, 16(2):151–195, 1991.

3. Charles Consel and Olivier Danvy. For a better support of static data flow. In John Hughes, editor, Proc. Functional Programming Languages and Computer Architecture 1991, number 523 in Lecture Notes in Computer Science, pages 496–519, Cambridge, MA, 1991. Springer-Verlag.

4. Charles Consel and Olivier Danvy. Tutorial notes on partial evaluation. In Proceedings of the 1993 ACM SIGPLAN Symposium on Principles of Programming Languages, pages 493–501, Charleston, South Carolina, January 1993. ACM Press.

5. Olivier Danvy. Type-directed partial evaluation. In Proceedings of the 1996 ACM SIGPLAN Symposium on Principles of Programming Languages, pages 242–257, St. Petersburg, Fla., January 1996. ACM Press.

6. Olivier Danvy and Andrzej Filinski. Representing control: A study of the CPS transformation. Mathematical Structures in Computer Science, 2:361–391, 1992.

7. Olivier Danvy, Karoline Malmkjær, and Jens Palsberg. Eta-expansion does The Trick. ACM Transactions on Programming Languages and Systems, 18(6):730–751, November 1996.

8. Dirk Dussart, Eddy Bevers, and Karel De Vlaminck. Polyvariant constructor specialization. In William Scherlis, editor, Proc. ACM SIGPLAN Symposium on Partial Evaluation and Semantics-Based Program Manipulation PEPM ’95, pages 54–63, La Jolla, CA, June 1995. ACM Press.

9. Dirk Dussart, Fritz Henglein, and Christian Mossin. Polymorphic recursion and subtype qualifications: Polymorphic binding-time analysis in polynomial time. In Alan Mycroft, editor, Proceedings of the 1995 International Static Analysis Symposium, number 983 in Lecture Notes in Computer Science, pages 118–136, Glasgow, Scotland, September 1995. Springer-Verlag.

10. Andrzej Filinski. Representing layered monads. In Alexander Aiken, editor, Proceedings of the 1999 ACM SIGPLAN Symposium on Principles of Programming Languages, pages 175–188, San Antonio, Texas, USA, January 1999. ACM Press.

11. Andrzej Filinski. Normalization by evaluation for the computational lambdacalculus. In Samson Abramsky, editor, Proc. of 5th Int. Conf. on Typed Lambda Calculi and Applications, TLCA’01, number 2044 in Lecture Notes in Computer Science, pages 151–165, Krakow, Poland, 2001. Springer-Verlag.

12. Haskell 98, a non-strict, purely functional language. http://www.haskell.org/ definition, December 1998.

13. John Hatcliff and Olivier Danvy. A computational formalization for partial evaluation. Mathematical Structures in Computer Science, 7(5):507–542, 1997.

14. John Hatcliff, Torben Æ. Mogensen, and Peter Thiemann, editors. Partial Evaluation—Practice and Theory. Proceedings of the 1998 DIKU International Summerschool, number 1706 in Lecture Notes in Computer Science, Copenhagen, Denmark, 1999. Springer-Verlag.

15. Neil Jones, Carsten Gomard, and Peter Sestoft. Partial Evaluation and Automatic Program Generation. Prentice-Hall, 1993.

16. Julia Lawall and Olivier Danvy. Continuation-based partial evaluation. In Proceedings of the 1994 ACM Conference on Lisp and Functional Programming, pages 227–238, Orlando, Florida, USA, June 1994. ACM Press.

17. Julia Lawall and Peter Thiemann. Sound specialization in the presence of computational effects. In Proceedings of the Theoretical Aspects of Computer Software, number 1281 in Lecture Notes in Computer Science, pages 165–190, Sendai, Japan, September 1997. Springer-Verlag.

18. John M. Lucassen and David K. Gifford. Polymorphic effect systems. In Proc. 15th Annual ACM Symposium on Principles of Programming Languages, pages 47–57, San Diego, California, January 1988. ACM Press.

19. John Mitchell. Foundations for Programming Languages. MIT Press, 1996. 20. Torben Æ. Mogensen. Inherited limits. In Hatcliff et al. [14], pages 189–202. 21. Eugenio Moggi. Notions of computations and monads. Information and Computation, 93:55–92, 1991.

22. Flemming Nielson, Hanne Riis Nielson, and Chris Hankin. Principles of Program Analysis. Springer Verlag, 1999.

23. Eijiro Sumii and Naoki Kobayashi. A hybrid approach to online and offline partial evaluation. Higher-Order and Symbolic Computation, 14(2/3):101–142, 2001.

24. Peter Thiemann. Combinators for program generation. Journal of Functional Programming, 9(5):483–525, September 1999.

25. Philip Wadler and Peter Thiemann. The marriage of monads and effects. ACM Transactions on Computational Logic, 4(1):1–32, January 2003.

26. Philip L. Wadler. Comprehending monads. In Proceedings of the 1990 ACM Conference on Lisp and Functional Programming, pages 61–78, Nice, France, 1990. ACM Press.

APPENDIX

A Proof of Lemma 1

The proofs are type checked Haskell scripts using the operator =:= for equality.

(MLB)

-- bind_let for S
ls_bind_let_S v e ma f =
  mlet v e ma >>= f =:=
  L v e ma >>= f =:=
  L v e (ma >>= f) =:=
  mlet v e (ma >>= f)

-- bind_let for C
ls_bind_let_C v e ma f =
  mlet v e ma >>= f =:=
  C (\ c -> unC (mlet v e ma) (\ x -> unC (f x) c)) =:=
  C (\ c -> unC (C (\ c -> LetExp v e (unC ma c))) (\ x -> unC (f x) c)) =:=
  C (\ c -> (\ c -> LetExp v e (unC ma c)) (\ x -> unC (f x) c)) =:=
  C (\ c -> LetExp v e (unC ma (\ x -> unC (f x) c))) =:=
  C (\ c -> LetExp v e ((\ c -> unC ma (\ x -> unC (f x) c)) c)) =:=
  C (\ c -> LetExp v e (unC (C (\ c -> unC ma (\ x -> unC (f x) c))) c)) =:=
  mlet v e (C (\ c -> unC ma (\ x -> unC (f x) c))) =:=
  mlet v e (ma >>= f)

(MIB) Analogous to (MLB)

(MRR)

-- mreset (return e) == return e
mreset_return_S e =
  mreset (return e) =:= -- def return
  mreset (V e) =:= -- def sreset
  V e =:= -- def return
  return e

mreset_return_C e =
  mreset (return e) =:= -- def return
  mreset (C (\ c -> c e)) =:= -- def creset
  C (\ c -> c (unC (C (\ c -> c e)) id)) =:= -- unC/C
  C (\ c -> c ((\ c -> c e) id)) =:= -- beta
  C (\ c -> c e) =:= -- def return
  return e

(MRL)

-- mreset (mlet v1 e1 me) == mreset me >>= \ e -> return (LetExp v1 e1 e)
mreset_mlet_S v1 e1 me =
  mreset (mlet v1 e1 me) =:= -- def slet
  mreset (L v1 e1 me) =:= -- def sreset

  mreset me >>= \ e’ -> return (LetExp v1 e1 e’)

mreset_mlet_C v1 e1 me =
  mreset (mlet v1 e1 me) =:= -- def clet
  mreset (C (\ c -> LetExp v1 e1 (unC me c))) =:= -- def creset
  C (\ c -> c (unC (C (\ c -> LetExp v1 e1 (unC me c))) id)) =:= -- simplify
  C (\ c -> c ((\ c -> LetExp v1 e1 (unC me c)) id)) =:= -- simplify
  C (\ c -> c (LetExp v1 e1 (unC me id))) =:= -- simplify
  C (\ c -> (\ x -> c (LetExp v1 e1 x)) (unC me id)) =:= -- simplify
  C (\ c -> (\ c -> c (unC me id)) (\ x -> c (LetExp v1 e1 x))) =:= -- def >>=
  C (\ c -> unC (C (\ c -> c (unC me id))) (\ x -> unC ((\ e -> C (\ c -> c (LetExp v1 e1 e))) x) c)) =:= -- def >>=
  C (\ c -> c (unC me id)) >>= \ e -> C (\ c -> c (LetExp v1 e1 e)) =:= -- def creset
  mreset me >>= \ e -> C (\ c -> c (LetExp v1 e1 e)) =:= -- def return
  mreset me >>= \ e -> return (LetExp v1 e1 e)

(MRI)

-- mreset (mif e m1 m2) ==
-- mreset m1 >>= \ e1 -> mreset m2 >>= \ e2 -> return (IfExp e e1 e2)
mreset_mif_S e m1 m2 =
  mreset (mif e m1 m2) =:= -- def sif
  mreset (I e m1 m2) =:= -- def sreset
  mreset m1 >>= \ e1 -> mreset m2 >>= \ e2 -> return (IfExp e e1 e2)

mreset_mif_C e m1 m2 =
  mreset (mif e m1 m2) =:= -- def cif
  mreset (C (\ c -> IfExp e (unC m1 c) (unC m2 c))) =:= -- def creset
  C (\ c -> c (unC (C (\ c -> IfExp e (unC m1 c) (unC m2 c))) id)) =:= -- simplify
  C (\ c -> c ((\ c -> IfExp e (unC m1 c) (unC m2 c)) id)) =:= -- simplify
  C (\ c -> c (IfExp e (unC m1 id) (unC m2 id))) =:= -- beta
  C (\ c -> (\ e1 -> (\ e2 -> c (IfExp e e1 e2)) (unC m2 id)) (unC m1 id)) =:= -- beta
  C (\ c -> (\ c -> c (unC m1 id)) (\ e1 -> (\ c -> (\ c -> c (unC m2 id)) (\ e2 -> (\ c -> c (IfExp e e1 e2)) c)) c)) =:= -- unC/C
  C (\ c -> (\ c -> c (unC m1 id)) (\ e1 -> unC (C (\ c -> (\ c -> c (unC m2 id)) (\ e2 -> unC (C (\ c -> c (IfExp e e1 e2))) c))) c)) =:= -- beta
  C (\ c -> (\ c -> c (unC m1 id)) (\ e1 -> unC ((\ e1 -> C (\ c -> (\ c -> c (unC m2 id)) (\ e2 -> unC ((\ e2 -> C (\ c -> c (IfExp e e1 e2))) e2) c))) e1) c)) =:= -- alpha
  C (\ c -> (\ c -> c (unC m1 id)) (\ x -> unC ((\ e1 -> C (\ c -> (\ c -> c (unC m2 id)) (\ x -> unC ((\ e2 -> C (\ c -> c (IfExp e e1 e2))) x) c))) x) c)) =:= -- simplify unC/C
  C (\ c -> unC (C (\ c -> c (unC m1 id))) (\ e1 -> unC ((\ e1 -> C (\ c -> unC (C (\ c -> c (unC m2 id))) (\ e2 -> unC ((\ e2 -> C (\ c -> c (IfExp e e1 e2))) e2) c))) e1) c)) =:= -- def >>=
  C (\ c -> unC (C (\ c -> c (unC m1 id))) (\ e1 -> unC ((\ e1 -> C (\ c -> c (unC m2 id)) >>= \ e2 -> C (\ c -> c (IfExp e e1 e2))) e1) c)) =:= -- def >>=
  C (\ c -> c (unC m1 id)) >>= \ e1 -> C (\ c -> c (unC m2 id)) >>= \ e2 -> C (\ c -> c (IfExp e e1 e2)) =:= -- creset, return
  mreset m1 >>= \ e1 -> mreset m2 >>= \ e2 -> return (IfExp e e1 e2)

B Proof of Theorem 2

To avoid clutter in the proof, we assume that variable names are also expressions, that is, the type Var is a subset of the type Exp. This avoids wrapping variable names into the VarExp constructor and also extracting them from an expression. Also, we argue rather informally about the semantics, mostly using rewriting on the expression level. For the whole proof, we assume that the translated expression is interpreted in the S monad, which we have shown to be equivalent to the C monad.

Proof. By induction on the derivation of Γ ⊢ e : ρ ! t.

Case Γ, x : ρ ⊢ x : ρ ! 1. It holds that �return x�φ = V φ(x) ∈ �ρ ! 1� by assumption and definition of �·�.

Case Γ ⊢ c : (int, s) ! 1. It holds that �return c�φ = V c ∈ �(int, s) ! 1� since c ∈ �(int, s)�.

Case

Γ ⊢ e : (int, b) ! t Γ ⊢ succ^b e : (int, b) ! t

. There are two cases, b = s and b = d.

For b = s, we need to evaluate �T �succ^s e�φ = �T �e�>>=λv.return (succ v)�φ. By induction, �T �e�φ ∈ �(int, s) ! t�. An auxiliary induction on t shows that �T �e� >>= λv.return (succ v)�φ ∈ �(int, s) ! t�, too:

– Case 1. In this case, �T �e�φ = V v, for some v, and hence �V v >>= λv.return (succ v)�φ = �return (succ v)�φ[v → v] = V v.

– Case l(t^′). In this case, �T �e�φ = L x e v, for some v ∈ �(int, s) ! t^′�. Hence, by (MLB), �L x e v>>=λv.return (succ v)�φ = �L x e (v>>=λv.return (succ v))�φ = L x e �(v >>= λv.return (succ v))�φ. By the auxiliary induction, �(v >>= λv.return (succ v))�φ ∈ �(int, s)!t^′�, so that �L x e v>>=λv.return (succ v)�φ ∈ �(int, s) ! l(t^′)�.

– Case i(t^′, t^{′′}) is similar.

– Case t^′ ⊔ t^{′′} is immediate by appeal to the inductive hypothesis.

For b = d, we have to show that �T �succ^d e�φ = �T �e�>>=λe^′.return (succ^r e^′)�φ ∈ �(int, d) ! t� . The proof is analogous.

Case lift e is also analogous.

Case

Γ, x : ρ_2 ⊢ e : ρ_1 ! t ρ = (ρ_2^t−→ρ_1, b) ⊢wft ρ Γ ⊢ λ^bx.e : ρ ! 1 . There are two subcases.

Subcase b = s. Consider �T �λ^sx.e�φ = �return (λx.T �e�) = V �λx.T �e�φ. Clearly, this result matches the effect 1. Hence, it remains to show that �λx.T �e�φ ∈ �(ρ_2^t−→ρ_1, s)�. That is, for all v ∈ �ρ_2�, it must be that �T �e�φ[x := v] ∈ �ρ_1 !t�. But this is exactly the induction hypothesis.

Subcase b = d. Consider �T �λ^dx.e�φ = newVar>>=λx.T �e� >>= λe^′.return (λ^r x.e^′). Since �newVar�φ = V x, for some x ∈ �Var�, we can apply the induction hypothesis to find that �T �e�φ[x → VarExp x] ∈ �ρ_1 ! t�. Since ⊢wft ρ is assumed and b = d, we know that t = 1 and ρ_1 = (τ_1, d) and ρ_2 = (τ_2, d) (and fortunately, VarExp x ∈ �(τ_2, d)� = �Exp�), so that �ρ_1 ! t� = {V e | e ∈ �Exp�}. Again, by definition of >>=, the result of �T �λ^dx.e�φ is V (λ^r x.e^′), for some e^′ ∈ �Exp�. That is, it is an element of �ρ ! 1�.

Case

Γ ⊢ e_1 : (ρ_2^t−→ρ_1, b) ! t_1 Γ ⊢ e_2 : ρ_2 ! t_2 Γ ⊢ e_1@^be_2 : ρ_1 ! t_1 ≫ t_2 ≫ t

. Again, there are two subcases.

Subcase b = s. In this case, �T �e_1@^se_2�φ = �T �e_1�>>=λf.T �e_2� >>= λv.f@v�φ. By induction �T �e_1�φ ∈ �(ρ_2^t−→ρ_1, s) ! t_1� so that f ∈ �(ρ_2^t−→ρ_1, s)� = {f | (∀v ∈ �ρ_2�) f v ∈ �ρ_1 ! t�}. Also, by induction, �T �e_2�φ ∈ �ρ_2 ! t_2� so that v ∈ �ρ_2� by definition of >>=. Hence, �f@v�φ[f . . . , v . . .] ∈ �ρ_1 ! t�. An auxiliary induction on t_2 shows that �T �e_2�>>=λv.f@v�φ[f . . .] ∈ �ρ_1 !t_2 ≫ t� and another auxiliary induction on t_1 shows the claim �T �e_1� >>= λf.T �e_2� >>= λv.f@v�φ ∈ �ρ_1 ! t_1 ≫ t_2 ≫ t�.

Subcase b = d. In this case, �T �e_1@^de_2�φ = �T �e_1�>>=λe^′_1.T �e_2� >>= λe^′_2.return (e^′_1@^re^′_2)�φ. By Lemma 2, ⊢wft (ρ_2^t−→ρ_1, d) so that t = 1, ρ_2 = (τ_2, d), and ρ_1 = (τ_1, d). By

induction, �T �e_1�φ ∈ �(ρ_2^t−→ρ_1, d)!t_1�, which means (by definition of >>=) that e^′_1 ∈ �Exp�. Also by induction, �T �e_2�φ ∈ �ρ_2 ! t_2� which means (by definition of >>= and since ρ_2 = (τ_2, d)) that e^′_2 ∈ �Exp�. Hence, �return (e^′_1@^re^′_2)�φ[e^′_1 . . . e^′_2 . . .] ∈ �ρ_1 ! 1� = �ρ_1 ! t� and two auxiliary inductions on t_2 and t_1 yield �T �e_1� >>= λe^′_1.T �e_2� >>= λe^′_2.return (e^′_1@^re^′_2)�φ ∈ �ρ_1 ! t_1 ≫ t_2 ≫ t�, as required.

Case

Γ ⊢ e_1 : (τ, s) ! t_1 Γ, x : (τ, s) ⊢ e_2 : ρ_2 ! t_2 Γ ⊢ let^s x = e_1 in e_2 : ρ_2 ! t_1 ≫ t_2

. Analogous to considering

(λ^sx.e_2)@^se_1.

Case

Γ ⊢ e_1 : (τ, d) ! t_1 Γ, x : (τ, d) ⊢ e_2 : ρ_2 ! t_2 Γ ⊢ let^d x = e_1 in e_2 : ρ_2 ! t_1 ≫ l(t_2)

. Consider �T �let^d x = e_1 in e_2�φ =

�T �e_1� >>= λe^′_1.newVar >>= λx.mlet x e^′_1 T �e_2�φ. By induction, �T �e_1�φ ∈ �(τ, d) ! t_1� so that (by def of >>=) e^′_1 ∈ �Exp�. By assumption, newVar is equal to return x, for some fresh variable x. Since VarExp x ∈ �(τ, d)�, induction yields that �T �e_2�φ[e^′_1 . . . , x → VarExp x] ∈ �ρ_2 ! t_2�. By definition of mlet, �mlet x e^′_1 T �e_2�φ[e^′_1 . . . , x . . .] = L x e^′_1 v, for some v ∈ �ρ_2 ! t_2�. By definition that means L x e^′_1 v ∈ �ρ_2 ! l(t_2)�. An auxiliary induction on t_1 establishes that �T �e_1� >>= λe^′_1.newVar >>= λx.mlet x e^′_1 T �e_2�φ ∈ �ρ_2 ! t_1 ≫ l(t_2)�.

Case

Γ ⊢ e_1 : (int, s) ! t_1 Γ ⊢ e_2 : ρ ! t_2 Γ ⊢ e_3 : ρ ! t_3 Γ ⊢ if0^s e_1 e_2 e_3 : ρ ! t_1 ≫ (t_2 ⊔ t_3)

. Consider �T �if0^s e_1 e_2 e_3�φ =

�T �e_1� >>= λv_1.if0 v_1 T �e_2� T �e_3�φ. By induction �T �e_1�φ ∈ �(int, s) ! t_1� so that v_1 ∈ �(int, s)�.

Subcase v_1 = 0. In that case �if0 v_1 T �e_2� T �e_3�φ[x → 0] = �T �e_2�φ. By induction, �T �e_2�φ ∈ �ρ!t_2�, hence so is �if0 v_1 T �e_2� T �e_3�φ[x → 0] ∈ �ρ!t_2�. Since �ρ!t_2� ⊆ �ρ!t_2⊔t_3�, it holds that �if0 v_1 T �e_2� T �e_3�φ[x → 0] ∈ �ρ!t_2⊔t_3�, too. As usual, an auxiliary inducition on t_1 and the definition of >>= establish that �T �e_1� >>= λv_1.if0 v_1 T �e_2� T �e_3�φ ∈ �ρ ! t_1 ≫ (t_2 ⊔ t_3)�.

Subcase v_1 = v^′_1 + 1. Analogous.

Case

Γ ⊢ e_1 : (τ, d) ! t_1 Γ ⊢ e_2 : ρ ! t_2 Γ ⊢ e_3 : ρ ! t_3 Γ ⊢ if0^d e_1 e_2 e_3 : ρ ! t_1 ≫ i(t_2, t_3)

. Consider �T �if0^d e_1 e_2 e_3�φ =

�T �e_1� >>= λe^′_1.mif e^′_1 T �e_2� T �e_3�φ. By induction, �T �e_1�φ ∈ �(τ, d) ! t_1� so that e^′_1 ∈ �(τ, d)�. Further induction yields �T �e_2�φ ∈ �ρ ! t_2� and �T �e_3�φ ∈ �ρ ! t_3�. Hence �mif e^′_1 T �e_2� T �e_3�φ[e^′_1 . . .] = Ie^′_1 v_2 v_3 where v_2 ∈ �ρ ! t_2� and v_2 ∈ �ρ ! t_3�. Clearly, Ie^′_1 v_2 v_3 ∈ �ρ ! i(t_2, t_3)�. Finally, an auxiliary induction on t_1 yields that �T �e_1� >>= λe^′_1.mif e^′_1 T �e_2� T �e_3�φ ∈ �ρ ! t_1 ≫ i(t_2, t_3)�.

Case Γ ⊢ e : (τ, d) ! t Γ ⊢ reset e : (τ, d) ! 1

. Consider �T �reset e�φ = �mreset T �e�φ.

By induction, �T �e�φ ∈ �(τ, d) ! t�. An auxiliary induction on t establishes the claim: Let m ∈ �(τ, d) ! t�, then mreset m ∈ �(τ, d) ! 1�.

Subcase t = 1. In this case, m = V e, for some e ∈ �Exp�. By definition of

mreset, mreset m = V e ∈ �(τ, d) ! 1�.

Subcase t = l(t^′). In this case, m = L x e m^′, for some m^′ ∈ �(τ, d)!t^′�. Hence,

mreset m = mreset (L x e m^′) = mreset m^′ >>= λe^′.return (LetExp x e e^′).

By induction, mreset m^′ ∈ �(τ, d) ! 1� and hence it holds that (by def of >>=)

mreset m^′ >>= λe^′.return (LetExp x e e^′) ∈ �(τ, d) ! 1�.

Subcase t = i(t^′, t^{′′}). In this case, m = I e m^′ m^{′′}, for some m^′ ∈ �(τ, d) ! t^′� and m^{′′} ∈ �(τ, d) ! t^{′′}�. Hence, mreset m = mreset (I e m^′ m^{′′}) = mreset m^′ >>= λe^′.mreset m^{′′} >>= λe^{′′}.return (IfExp e e^′ e^{′′}). By induction, mreset m^′ ∈ �(τ, d) ! 1� and mreset m^{′′} ∈ �(τ, d) ! 1� and hence it holds that (by def of >>=) mreset m^′ >>= λe^′.mreset m^{′′} >>= λe^{′′}.return (IfExp e e^′ e^{′′}) ∈ �(τ, d) ! 1�.

Subcase t = t^′ ⊔ t^{′′}. In this case, m ∈ �(τ, d) ! t^′� or m ∈ �(τ, d) ! t^{′′}�. In both cases, induction is applicable and yields that mreset m ∈ �(τ, d) ! 1�.